

Homework of DSP

1. Compute the output of $y = 1+2+3+4+..+300$.
2. For an equation $y = ax^2+bx+c$ with its parameters $a = 2$, $b = 3$, $c = 1$, compute the output value of y using direct addressing, indirect addressing or MAC instruction method for the input value of $x = [-10,-9,-8,..,-2,-1,0,1,2,..,8,9,10]$.
3. For a filter represented by $y(n)=a_1x(n)+a_2x(n-1)+a_3x(n-2)+b_1y(n-1)$ with $a_1=0.2$, $a_2= -0.3$, $a_3= 0.4$, $b_1= 0.1$, find the top 15 data of the output sequence y using direct addressing, indirect addressing or MACD instruction method for the input sequences of $x = [0.1,0.2,0.3,0.4,0.5,0.6,0.7,0.8,0.9,1.0,0,0,0,0,0]$. Assume that the initial values in the filter are all 0.

Solution of Question 1

$$y = \sum_{n=0}^{300} X_n$$

```
*test1.asm
        .title      " y=1+2+..+100"
        .include    "pathway.inc"
        .entry
        .ds 0300h
XCOUNT  .word 0
Y100    .word 0
        .ps 0fe00h      ; starting address for this section is
B Phantom_ISR      ; (02h) Interrupt Level 1
B Phantom_ISR      ; (04h) Interrupt Level 2
B Phantom_ISR      ; (06h) Interrupt Level 3
B Phantom_ISR      ; (08h) Interrupt Level 4
B Phantom_ISR      ; (0Ah) Interrupt Level 5
        .ps 0fe50h
        .entry

start
        SETC      INTM      ; INTM = 1, disable global interrupts
        LDP       #0e8h      ; DP -> 0x7400 - 0x747f (Event Manager)
        SPLK      #0ffffh, IFR ; clear all pending interrupts
        SPLK      #0010h, IMR ; enable Level 5 interrupts for monitor
        CLRC      INTM      ; enable interrupts
        CLRC      SXM      ; disable sign extension
        LDP       #06        ; setting data page
        LAR       AR2,#299
        MAR       *,AR2

loop
        ZAC
        LACC      XCOUNT
        ADD       #1
        SACL      XCOUNT
        ADD       Y100
        SACL      Y100
        BANZ     loop

nn      NOP
        B        nn

Phantom_ISR:
        B        Phantom_ISR
        .end
```

Solution of Question 2

```

; filename: test2.asm
        .title      " y=ax^2+bx+c for x=-10~10 "
        .include    "pathway.inc"
        .entry
        .ds 0300h

XN      .word      -10,-9,-8,-7,-6,-5,-4,-3,-2,-1,0,1,2,3,4,5,6,7,8,9,10
A       .word      2
B       .word      3
C       .word      1
X2      .word      0
X1      .word      0
Y       .word      0

        .ps 0fe00h      ; starting address for this section is
                        ; 0fe00h in Program Space

        B 0000h        ; (00h) Hardware Reset
        B Phantom_ISR ; (02h) Interrupt Level 1
        B Phantom_ISR ; (04h) Interrupt Level 2
        B Phantom_ISR ; (06h) Interrupt Level 3
        B Phantom_ISR ; (08h) Interrupt Level 4
        B Phantom_ISR ; (0Ah) Interrupt Level 5

        .ps 0fe50h
        .entry

start:
        SETC  INTM      ; INTM = 1, disable global interrupts
        LDP   #0e8h     ; DP -> 0x7400 - 0x747f (Event Manager)
        SPLK #0ffffh, IFR ; clear all pending interrupts
        SPLK #0010h, IMR ; enable Level 5 interrupts
                        ; so monitor can communicate with host

        CLRC  INTM      ; enable interrupts
        CLRC  SXM       ; disable sign extension

        LDP   #06H
        LAR   AR3,#14H
        LAR   AR2,#Y
        LAR   AR1,#XN
    
```

```

loop:
        MAR   *,AR1
        LACC  *+
        SACL  X1      ; X1 = XN
        LT   X1
        MPY   X1
        PAC
        SACL  X2      ; X2 = X*X
        ZAC          ; Acc = 0
        LT   X2      ; Treg = X^2
        MPY   A       ; Preg = A*X^2
        APAC          ; Acc = A*X^2
        LT   X1      ; Treg = X
        MPY   B       ; Preg = B*X
        APAC          ; Acc = A*X^2+B*X
        ADD   C       ; Acc = A*X^2+B*X+C
        MAR   *,AR2
        SACL  *+
        MAR   *,AR3
        BANZ  loop

nn:
        NOP
        NOP
        NOP
        B     nn

Phantom_ISR:
        B     Phantom_ISR
        .end
    
```

Solution of Question 3

```

; filename: test3.asm
.title      "y(n)=A1*x(n)+A2*x(n-1)+A3*x(n-2)+B1*y(n-1)"
.include    "pathway.inc"
.entry
.ds 0300h

XN          .word  3277,6554,9830,13107,16384,19661,22938,26214,29491,32767
            .word  0,0,0,0,0,0
; Xn = [0.1,0.2,0.3,0.4,0.5,0.6,0.7,0.8,0.9,1,0,0,0,0,0];
X0          .word  0
X1          .word  0
X2          .word  0
Y1          .word  0
A0          .word  6554          ; 0.2
A1          .word -9830          ; -0.3
A2          .word 13107         ; 0.4
B1          .word 3277          ; 0.1
ZEO         .word 0
Y           .word 0
            .ps 0fe00h          ; starting address for this section is
                                ; 0fe00h in Program Space
            b 0000h            ; (00h) Hardware Reset
            b Phantom_ISR      ; (02h) Interrupt Level 1
            b Phantom_ISR      ; (04h) Interrupt Level 2
            b Phantom_ISR      ; (06h) Interrupt Level 3
            b Phantom_ISR      ; (08h) Interrupt Level 4
            b Phantom_ISR      ; (0Ah) Interrupt Level 5

            .ps 0fe50h

            .entry
start:
SETC        INTM                ; INTM = 1, disable global interrupts
LDP         #0e8h                ; DP -> 0x7400 - 0x747f (Event Manager)
SPLK        #0ffffh, IFR         ; clear all pending interrupts
SPLK        #0010h, IMR          ; enable Level 5 interrupts
                                ; so monitor can communicate with host
CLRC        INTM                ; enable interrupts

```

```

CLRC        SXM                  ; disable sign extension

LDP         #06H
LAR         AR3,#0EH
LAR         AR2,#Y
LAR         AR1,#XN
SETC        SXM

loop:
LAR         AR6,#A2
LAR         AR5,#X2
ZAC                     ; ACC=0
LT          ZEO
MPY         ZEO            ; P=0
MAR         *,AR1
LACC        *+
SACL        X0            ; X0 = Xn
MAR         *,AR5
LT          *- ,AR6        ; T=X2
MPY         *- ,AR5        ; P=X2*A2
LTD         *- ,AR6        ; ACC=X2*A2          T=X1   X2=X1
MPY         *- ,AR5        ; P=X1*A1
LTD         *- ,AR6        ; ACC=X2*A2+X1*A1   T=X0   X1=X0
MPY         *- ,AR5        ; P=X0*A0
APAC                     ; ACC=X2*A2+X1*A1+X0*A0
LT          Y1            ; T=Y1
MPY         B1            ; P=B1*Y1
APAC                     ; ACC=X2*A2+X1*A1+X0*A0+B1*Y1
SACH        Y1,1          ; Y1=MSB(Shift(ACC,1))
MAR         *,AR2
SACH        *+,1,AR3      ; Y=MSB(Shift(ACC,1))
BANZ        loop

nn:
NOP
B          nn

Phantom_ISR:
B          Phantom_ISR
.end

```